PROC FORMAT – Not Just Another Pretty Face
Lois Levin, Independent Consultant, Bethesda, MD

ABSTRACT

PROC FORMAT, because of its name, is most often used to change the appearance of data for presentation. But PROC FORMAT can be used for much more than the cosmetic transformation of data. It can be a tool for redefining data, recoding data, transforming data for tabulation and reporting, subsetting large data files and more. This paper will present a few applications of PROC FORMAT that go beyond the basic display of data and show that PROC FORMAT has real character that goes deeper than just its pretty face.

INTRODUCTION

There is a wealth of stored formats in SAS® that may be used to express data in many ways. PROC FORMAT provides a method of going beyond those supplied methods to transform data in any way you choose. This paper is not meant to be a tutorial on the use of PROC FORMAT with all of its options and keywords. Rather, it is a collection of interesting applications of PROC FORMAT that will allow you to control the expression and management of data.

The examples included will show how PROC FORMAT can be used to:

1. Group/recode data
2. Identify missing data
3. Display the same variable several ways
4. Perform a table lookup
5. Display negative percent values with negative signs and % signs.
6. Display datetimes
7. Code styles conditionally
8. Extract records from a very large dataset

1 – GROUP/RECODE DATA

PROC FORMAT can be used to define groups of data. If your data are numeric, the groups allow you to report the data as meaningful distributions. The groups also allow data to be recoded based on the group definitions. Groups can be defined and then codes assigned to each group. This could also be accomplished with a sequence of IF statements, but the format will probably take less code and can often run faster. The transformation using one PUT statement is much faster than many IF statements.

In the following example we want to report temperature data by groups of “Low”, “Medium”, “High”. We will use 2 formats -- one to recode the data and one to assign the names. Why not just recode the temperature data as “Low”, “Medium”, or “High”? If we had a large, permanent dataset, we would have to store all of those literal descriptions for every observation. It is much more efficient to recode the data to short character values (Short character values take up less space than standard numeric values), and then use the literals for reporting only. The other reason for using two formats is that the formats will be sorted alphabetically. So our report would show the data as “High”, “Low”, and “Medium” in that order. By coding them in sequence first, we can get them in the order we like.
Example 1 – Program

```
proc format;
  value tempfmt   low < 61 = ' 1'
                   61 <= 63 = ' 2'
                   63 <= high = ' 3'
                   other = '  ';
  value $ codefmt
       '1'='Low   '
       '2'='Medium'
       '3'='High  ';
run;

data pme;
  set pme;
  *
  --- replace list of IF statements ---*;
  * if avgtemp < 61 then code=' 1';
  * else if avgtemp < 63 then code=' 2';
  * else if avgtemp <999 then code=' 3';
  * else code='  ';
  *-------------------------------------------------------------*;
  tempcode=put(avgtemp, tempfmt.);
run;

proc freq data=pme;
  tables tempcode/missing;
  format tempcode $codefmt.;
run;
```

Example 1 – Output

```
Example 1 - Group/Recode Data
Temperatures in Portland, ME - September 2010

<table>
<thead>
<tr>
<th>tempcode</th>
<th>Frequency</th>
<th>Percent</th>
<th>Cumulative Frequency</th>
<th>Cumulative Percent</th>
</tr>
</thead>
<tbody>
<tr>
<td>Low</td>
<td>11</td>
<td>36.67</td>
<td>11</td>
<td>36.67</td>
</tr>
<tr>
<td>Medium</td>
<td>6</td>
<td>20.00</td>
<td>17</td>
<td>56.67</td>
</tr>
<tr>
<td>High</td>
<td>13</td>
<td>43.33</td>
<td>30</td>
<td>100.00</td>
</tr>
</tbody>
</table>
```
2 – IDENTIFY MISSING DATA

A specific case of using formats to group data is this one which divides the data into just 2 groups, missing and non-missing. It is a convenient way to take a quick initial look at your data. This example defines 2 formats – one numeric and one character because we have to distinguish between the numeric missing (.) and the character missing ( ). The 2 formats can be applied to all of the variables in a dataset.

Example 2 - Program

```sas
proc format;
  value $ missfmt ' '="Missing"
      other="Not Missing"
  ;
  value nmissfmt . ="Missing"
      other="Not Missing"
  ;
run;

proc freq data=home.ex2;
  tables _numeric_ _character_/missing;
  format _numeric_ nmissfmt._character_ $missfmt. ;
run;
```

Example 2 – Output

```
Example 2 - Identify Missing Data

Percent 1

<table>
<thead>
<tr>
<th>Percent1</th>
<th>Frequency</th>
<th>Percent</th>
<th>Cumulative</th>
<th>Cumulative</th>
</tr>
</thead>
<tbody>
<tr>
<td>Missing</td>
<td>160</td>
<td>1.60</td>
<td>160</td>
<td>1.60</td>
</tr>
<tr>
<td>Not Missing</td>
<td>9840</td>
<td>98.40</td>
<td>10000</td>
<td>100.00</td>
</tr>
</tbody>
</table>

Amount 2

<table>
<thead>
<tr>
<th>Amount2</th>
<th>Frequency</th>
<th>Percent</th>
<th>Cumulative</th>
<th>Cumulative</th>
</tr>
</thead>
<tbody>
<tr>
<td>Not Missing</td>
<td>10000</td>
<td>100.00</td>
<td>10000</td>
<td>100.00</td>
</tr>
</tbody>
</table>

Date 1

<table>
<thead>
<tr>
<th>Date1</th>
<th>Frequency</th>
<th>Percent</th>
<th>Cumulative</th>
<th>Cumulative</th>
</tr>
</thead>
<tbody>
<tr>
<td>Missing</td>
<td>10000</td>
<td>100.00</td>
<td>10000</td>
<td>100.00</td>
</tr>
</tbody>
</table>
```
DISPLAY THE SAME VARIABLE SEVERAL WAYS

Here is a case where we have one date variable but we want to display the data in several ways. The format **ex3fmt** specifies the date as a 4-digit year, a month/year, and also a specific day like 01JAN2010.

This example allows us to see older dates by year, recent dates by month/year, and current dates as a specific day. We are using the technique of nested formats, meaning that our format references another format in its definition. The referenced format is coded with brackets (or parentheses and vertical bars, e.g. |(year4.|)) to indicate that it is defined elsewhere. In this case it is just a standard SAS format but it could reference another user-defined format.

Note that the ranges of the dates do not overlap. If you wanted to see different displays of the same data, you could use the MULTILABEL option in PROC FORMAT to define multiple formats for a given date range.

Example 3 - Program

```sas
proc format;
  value ex3fmt  low='31DEC2008' d=[year4.]
                   '01JAN2009' d='31DEC2009' d=[monyy7.]
                   '01JAN2010' d-high=[date9.];
run;

title 'Example 3 - Display One Variable Several Ways';
proc freq data=ex3;
  tables ex3dat;
  format ex3dat ex3fmt.;
run;
```

Example 3 - Display One Variable Several Ways

<table>
<thead>
<tr>
<th>ex3dat</th>
<th>Frequency</th>
<th>Percent</th>
<th>Cumulative Frequency</th>
<th>Cumulative Percent</th>
</tr>
</thead>
<tbody>
<tr>
<td>2005</td>
<td>749</td>
<td>19.51</td>
<td>749</td>
<td>19.51</td>
</tr>
<tr>
<td>2006</td>
<td>807</td>
<td>21.02</td>
<td>1556</td>
<td>40.52</td>
</tr>
<tr>
<td>2007</td>
<td>679</td>
<td>17.68</td>
<td>2235</td>
<td>58.20</td>
</tr>
<tr>
<td>2008</td>
<td>649</td>
<td>16.90</td>
<td>2884</td>
<td>75.10</td>
</tr>
<tr>
<td>JAN2009</td>
<td>59</td>
<td>1.54</td>
<td>2943</td>
<td>76.64</td>
</tr>
<tr>
<td>FEB2009</td>
<td>36</td>
<td>0.94</td>
<td>2979</td>
<td>77.58</td>
</tr>
<tr>
<td>MAR2009</td>
<td>52</td>
<td>1.35</td>
<td>3031</td>
<td>78.93</td>
</tr>
<tr>
<td>APR2009</td>
<td>56</td>
<td>1.46</td>
<td>3087</td>
<td>80.39</td>
</tr>
<tr>
<td>MAY2009</td>
<td>59</td>
<td>1.54</td>
<td>3146</td>
<td>81.93</td>
</tr>
<tr>
<td>JUN2009</td>
<td>56</td>
<td>1.46</td>
<td>3202</td>
<td>83.39</td>
</tr>
<tr>
<td>JUL2009</td>
<td>64</td>
<td>1.67</td>
<td>3266</td>
<td>85.05</td>
</tr>
<tr>
<td>AUG2009</td>
<td>44</td>
<td>1.15</td>
<td>3310</td>
<td>86.20</td>
</tr>
<tr>
<td>SEP2009</td>
<td>45</td>
<td>1.17</td>
<td>3355</td>
<td>87.37</td>
</tr>
<tr>
<td>OCT2009</td>
<td>59</td>
<td>1.54</td>
<td>3414</td>
<td>88.91</td>
</tr>
<tr>
<td>NOV2009</td>
<td>53</td>
<td>1.38</td>
<td>3467</td>
<td>90.29</td>
</tr>
<tr>
<td>DEC2009</td>
<td>111</td>
<td>2.89</td>
<td>3578</td>
<td>93.18</td>
</tr>
<tr>
<td>02JAN2010</td>
<td>16</td>
<td>0.42</td>
<td>3594</td>
<td>93.59</td>
</tr>
<tr>
<td>30JAN2010</td>
<td>29</td>
<td>0.76</td>
<td>3623</td>
<td>94.35</td>
</tr>
<tr>
<td>01FEB2010</td>
<td>32</td>
<td>0.83</td>
<td>3655</td>
<td>95.18</td>
</tr>
<tr>
<td>28FEB2010</td>
<td>27</td>
<td>0.70</td>
<td>3682</td>
<td>95.89</td>
</tr>
<tr>
<td>01MAR2010</td>
<td>24</td>
<td>0.63</td>
<td>3706</td>
<td>96.51</td>
</tr>
<tr>
<td>31MAR2010</td>
<td>38</td>
<td>0.99</td>
<td>3744</td>
<td>97.50</td>
</tr>
<tr>
<td>01APR2010</td>
<td>31</td>
<td>0.81</td>
<td>3775</td>
<td>98.31</td>
</tr>
<tr>
<td>30APR2010</td>
<td>64</td>
<td>1.67</td>
<td>3839</td>
<td>99.97</td>
</tr>
<tr>
<td>07MAY2010</td>
<td>1</td>
<td>0.03</td>
<td>3840</td>
<td>100.00</td>
</tr>
</tbody>
</table>
4 – PERFORM A TABLE LOOKUP

One very common and useful application of PROC FORMAT is for table lookups. You can have data in one file and some related data in another file and you can combine them based on a common variable. This type of task is often done with a DATA step and a MERGE statement. But if we do it with formats, the PUT statement that defines the new variable will use much less CPU than a MERGE statement.

In this example we have one dataset with SAILNO and NAME and another with race results by SAILNO. We want a report of race results with the SAILNO and NAME. We are going to do it by creating a format using the SAILNO and use it to lookup the NAME when we read in the race data.

There are several ways to define the format. A VALUE statement could be explicitly defined. If the list of values is long, the data can be read in with a DATA step and a format generated from the data using the CNTLIN option. In this example we have a relatively short list of values (for SAILNO and NAME) so we will create the format by inputting the data into a macro variable and then use that macro variable to define the format value. The macro variable NAMELIST is created by using PROC SQL with the SELECT…INTO: construct to place the list of values of the variable SAILNO into the macro variable. Then the format is defined by referring to the macro variable &NAMELIST. The new variable BOATNAME is created by formatting the variable SAILNO as $NAMEFMT.

The excerpt from the log shows how the macro variable is interpreted to include the VALUE definition. The report shows the new variable BOATNAME which has been created by using the variable SAILNO with the newly defined format NAMEFMT. It performs the equivalent of:

data ex4;
  merge sailname saildat1;
  by sailno;
run;

Example 4 – Data

data saildat1;
  input position sailno $5. race1 race2 race3 race4 race5 race6 race7 race8;
cards;
  1 42968 3 1 4 1 1 1 1
  2 40738 4 4 1 3 2 6 2 7
  3 73000 1 3 3 5 10 2 6 3
  4 32339 2 2 5 9 8 3 3 2
  5 53427 8 8 6 2 4 4 8 4
  6 97864 6 5 7 6 5 9 4 5
  7 33751 7 6 10 5 6 7 5 8
  8 41870 9 9 7 3 5 10 9
  9 40248 5 7 2 8 11 11 8 13
  10 73027 11 11 8 10 7 10 9 10
  11 73425 10 10 11 11 9 8 11 7
  12 41791 13 13 13 12 12 13 13 13
run;

data sailname;
  input sailno $5. sailname $20.;
cards;
  42968 Mr Bill's Wild Ride
  40738 Maggie
  73000 Medicine Man
  32339 Bump in the Night
  53427 Windependent
  97864 Dakota Girl
  33751 Bzing
  41870 T-Bone
  40248 Bad Company
  73027 Eighth Deadly Sin
  73425 Rebel Yell
  41791 Fatal Attraction
run;
Example 4 – Program

options symbolgen;

data saildat1;
  set saildat1;
  total = sum(of race1-race8);
run;

data sailname;
  set sailname;
  qt='"';
  eq='=';
run;

proc sql noprint;
  select cats(
    qt,sailno,qt,eq,qt,sailname,qt)
  into :namelist separated by ','
  from sailname;
quit;

proc format;
  value $ namefmt &namelist;
run;

data ex4;
  set saildat1;
    Boatname=put(sailno,$namefmt.);
run;

Example 4 – Log

60
61   proc format;
62 value $ namefmt &namelist;
SYMBOLGEN:  Macro variable NAMELIST resolves to "42968"="Mr Bill's Wild Ride"
            "40738"="Maggie" "73000"="Medicine Man" "32339"="Bump in the Night"
            "53427"="Winependent" "97864"="Dakota Girl" "33751"="Bzing"
            "41870"="T-Bone" "40248"="Bad Company" "73027"="Eighth Deadly Sin"
            "73425"="Rebel Yell" "41791"="Fatal Attraction"
NOTE: Format $NAMEFMT has been output.
63 run;
Example 4 – Output

Using formats created from data
Race Results - J35 Class - Annapolis Race Week 2010

<table>
<thead>
<tr>
<th>position</th>
<th>Boatname</th>
<th>sailno</th>
<th>race1</th>
<th>race2</th>
<th>race3</th>
<th>race4</th>
<th>race5</th>
<th>race6</th>
<th>race7</th>
<th>race8</th>
<th>total</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>Mr Bill’s Wild Ride</td>
<td>42968</td>
<td>3</td>
<td>1</td>
<td>4</td>
<td>1</td>
<td>1</td>
<td>1</td>
<td>1</td>
<td>1</td>
<td>13</td>
</tr>
<tr>
<td>2</td>
<td>Maggie</td>
<td>40738</td>
<td>4</td>
<td>4</td>
<td>1</td>
<td>3</td>
<td>2</td>
<td>6</td>
<td>2</td>
<td>7</td>
<td>29</td>
</tr>
<tr>
<td>3</td>
<td>Medicine Man</td>
<td>73000</td>
<td>1</td>
<td>3</td>
<td>3</td>
<td>5</td>
<td>10</td>
<td>2</td>
<td>6</td>
<td>3</td>
<td>33</td>
</tr>
<tr>
<td>4</td>
<td>Bump in the Night</td>
<td>32339</td>
<td>2</td>
<td>2</td>
<td>5</td>
<td>9</td>
<td>8</td>
<td>3</td>
<td>3</td>
<td>2</td>
<td>34</td>
</tr>
<tr>
<td>5</td>
<td>Windependent</td>
<td>53427</td>
<td>8</td>
<td>8</td>
<td>6</td>
<td>2</td>
<td>4</td>
<td>4</td>
<td>8</td>
<td>4</td>
<td>44</td>
</tr>
<tr>
<td>6</td>
<td>Dakota Girl</td>
<td>97864</td>
<td>6</td>
<td>5</td>
<td>7</td>
<td>6</td>
<td>5</td>
<td>9</td>
<td>4</td>
<td>5</td>
<td>47</td>
</tr>
<tr>
<td>7</td>
<td>Bzing</td>
<td>33751</td>
<td>7</td>
<td>6</td>
<td>10</td>
<td>5</td>
<td>6</td>
<td>7</td>
<td>5</td>
<td>8</td>
<td>54</td>
</tr>
<tr>
<td>8</td>
<td>T-Bone</td>
<td>41870</td>
<td>9</td>
<td>9</td>
<td>9</td>
<td>7</td>
<td>3</td>
<td>5</td>
<td>10</td>
<td>9</td>
<td>61</td>
</tr>
<tr>
<td>9</td>
<td>Bad Company</td>
<td>40248</td>
<td>5</td>
<td>7</td>
<td>2</td>
<td>8</td>
<td>11</td>
<td>11</td>
<td>8</td>
<td>13</td>
<td>65</td>
</tr>
<tr>
<td>10</td>
<td>Eighth Deadly Sin</td>
<td>73027</td>
<td>11</td>
<td>11</td>
<td>8</td>
<td>10</td>
<td>7</td>
<td>10</td>
<td>9</td>
<td>10</td>
<td>76</td>
</tr>
<tr>
<td>11</td>
<td>Rebel Yell</td>
<td>73425</td>
<td>10</td>
<td>10</td>
<td>11</td>
<td>11</td>
<td>9</td>
<td>8</td>
<td>11</td>
<td>7</td>
<td>77</td>
</tr>
<tr>
<td>12</td>
<td>Fatal Attraction</td>
<td>41791</td>
<td>13</td>
<td>13</td>
<td>13</td>
<td>12</td>
<td>13</td>
<td>13</td>
<td>13</td>
<td>13</td>
<td>102</td>
</tr>
</tbody>
</table>

5 – DISPLAY NEGATIVE PERCENT VALUES WITH NEGATIVE SIGNS AND % SIGNS

SAS has many built-in formats that you can use to display data. But a finite list is never enough so PROC FORMAT supplies the picture format capability so that you can design your own format. Where the VALUE statement lets you display data as formatted numerics or literals, the PICTURE statement lets you display numeric data in special patterns using special characters. The SSN and phone number pictures are the most obvious examples of picture formats. They use numbers with dashes in specific patterns. Now consider the percentw.d format. It displays numerics as percentages with the percent sign (%) and negative values within parentheses. In version 9, we have the percentn format that displays percentages with the percent sign (%) but negative numbers with a negative sign. Let’s build the picture format that is equivalent to the percentn format. The program will read in some data and print it with no format, so that you can see the raw values, with the percentw.d format, with the percentn format and with our very own picture format.

To create the format, we first have to convert the value to a percent and that is what the MULT option does. It moves the decimal point over to the place where we want to display it. Then we need to show the negative numbers with the negative sign and the PREFIX option defines this. So we need to define our format with two kinds of values – the negatives and everything else. The picture itself shows the layout of the number including the decimal point and the percent sign. The 9’s indicate that we want a value displayed in that position, even if it is zero. The 0’s indicate that those leading zeros may be dropped. If we used ’000.99’, then a value of .59 would be displayed as 59%. We have also used the ROUND option to ensure that the value is rounded, not truncated, to our designated 2 decimal places.

Example 5 - Program

```sas
proc format;
    picture pctpic (round) low=<0  ='0009.99%'  (prefix='-' mult=10000)  
        0-high='0009.99%'  (mult=10000) ;
run;

data t1;
    input @1 typenum $char5. @9 var1 10.6 ;
cards;
Type1  -0.005855
Type2   0.012498
Type3  -0.047033
Type4   -1. ;
run;
```
proc print data=t1 noobs split='*';
var typenum var1;
   label typenum="Type";;
   var1="Var1";;
run;

title 'Report without Format';
run;

proc print data=t1 noobs split='*';
var typenum var1;
   label typenum="Type";;
   var1="Var1";;
   format var1 percent9.2;
run;

title 'Report with Percent Format';
run;

proc print data=t1 noobs split='*';
var typenum var1;
   label typenum="Type";;
   var1="Var1";;
   format var1 percentn9.2;
run;

title 'Report with PercentN Format';
run;

proc print data=t1 noobs split='*';
var typenum var1;
   label typenum="Type";;
   var1="Var1";;
   format var1 pctpic.;
run;

title 'Report with Picture Format';
run;

Example 5 - Output

Report without Format

<table>
<thead>
<tr>
<th>Type</th>
<th>Var1</th>
</tr>
</thead>
<tbody>
<tr>
<td>-------</td>
<td>-------</td>
</tr>
<tr>
<td>Type1</td>
<td>-0.00586</td>
</tr>
<tr>
<td>Type2</td>
<td>0.01250</td>
</tr>
<tr>
<td>Type3</td>
<td>-0.04703</td>
</tr>
<tr>
<td>Type4</td>
<td>-1.00000</td>
</tr>
</tbody>
</table>

Report with Percent Format

<table>
<thead>
<tr>
<th>Type</th>
<th>Var1</th>
</tr>
</thead>
<tbody>
<tr>
<td>-------</td>
<td>--------</td>
</tr>
<tr>
<td>Type1</td>
<td>( 0.59%)</td>
</tr>
<tr>
<td>Type2</td>
<td>1.25%</td>
</tr>
<tr>
<td>Type3</td>
<td>( 4.70%)</td>
</tr>
<tr>
<td>Type4</td>
<td>(100.00%)</td>
</tr>
</tbody>
</table>
Report with PercentN Format

<table>
<thead>
<tr>
<th>Type</th>
<th>Var1</th>
</tr>
</thead>
<tbody>
<tr>
<td>Type1</td>
<td>-0.59%</td>
</tr>
<tr>
<td>Type2</td>
<td>1.25%</td>
</tr>
<tr>
<td>Type3</td>
<td>-4.70%</td>
</tr>
<tr>
<td>Type4</td>
<td>-100.00%</td>
</tr>
</tbody>
</table>

Report with Picture Format

<table>
<thead>
<tr>
<th>Type</th>
<th>Var1</th>
</tr>
</thead>
<tbody>
<tr>
<td>Type1</td>
<td>-0.59%</td>
</tr>
<tr>
<td>Type2</td>
<td>1.25%</td>
</tr>
<tr>
<td>Type3</td>
<td>-4.70%</td>
</tr>
<tr>
<td>Type4</td>
<td>-100.00%</td>
</tr>
</tbody>
</table>

6 – DISPLAY DATETIMES

There aren't too many stored datetime formats but the picture format with directives can be used to describe any picture for your date, time, or datetime variable. This technique is especially useful for formatting SAS dates as DBMS dates for use in SQL queries and for special reporting specifications. In this example we want to create one datetime as the current datetime and another as an assigned value. Both need to be formatted in this combination with dashes and dots. We are using the option DATATYPE to indicate that directives will be used to create a template for the picture format. The directives here are the % indicators for the components of the datetime. Since the result is a picture, we must convert the original numeric datetime value to a character value using the PUT function.

Example 6 – Program

```sas
options nodate ls=80;
proc format;
    picture dtpic
        other='%Y-%m-%d-%H.%M.%S' (datatype=datetime)
    ;
run;

data ex6;
    dt_now_standard = put('14Sep2011-08.00.00'dt,datetime16.);
    dt_now_picture  = put('14Sep2011-08.00.00'dt,dtpic.);
run;

title 'Example 6 – Display Datetimes';
proc print;
run;
```

Example 6 - Output

<table>
<thead>
<tr>
<th>Obs</th>
<th>dt_now_standard</th>
<th>dt_now_picture</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>14SEP11:08:00:00</td>
<td>2011-09-14-08:00.00</td>
</tr>
</tbody>
</table>
Suppose you are creating a report and you want to define a report style conditionally, depending on the value of the data. For example, you want to prepare an Excel spreadsheet with all the negative values in red. This is a very simple version of what is often called "traffic lighting" – using colors to code data in reports. You could do this in PROC REPORT using a compute block with an IF statement and a CALL DEFINE. You would have to create a compute block for every variable. But you could also use PROC FORMAT to define the style as a format and then just reference the format when you define the variable. In this case we are not going to reference the format in a format statement, but rather in a style specification. Here's the example.

The format defines a range of data and the color associated with each range. Of course you could define many more ranges and many more colors if you like. You could also use it to define a different kind of style, like a font, for instance. Then the style option in the DEFINE statement just references the format and you don't have to use any compute blocks at all.

Example 7 – Program

```plaintext
proc format;
   value negfmt
       low-0 = 'red'
       other = 'black'
   ;
run;

ods listing close;
ods html body="&home/ex7a.xls";

proc report data=ex7a nowd
    style(report)=[background=white foreground=black]
    style(column)=[background=white foreground=black font_size=2]
    style(header)=[background=white foreground=blue font_size=2]
    ;
columns code var1 var2 var3;
define code / 'Code';
define var1 / format=4.0 style=[foreground=negfmt.] 'Var 1';
define var2 / format=4.0 style=[foreground=negfmt.] 'Var 2';
define var3 / format=4.0 style=[foreground=negfmt.] 'Var 3';
run;

ods html close;
ods listing;
```

Example 7 – Output

The Excel report we created looks like this.

<table>
<thead>
<tr>
<th>Code</th>
<th>Var 1</th>
<th>Var 2</th>
<th>Var 3</th>
</tr>
</thead>
<tbody>
<tr>
<td>ABC</td>
<td>123</td>
<td>456</td>
<td>789</td>
</tr>
<tr>
<td>DEF</td>
<td>-321</td>
<td>345</td>
<td>654</td>
</tr>
<tr>
<td>GHI</td>
<td>567</td>
<td>-456</td>
<td>986</td>
</tr>
<tr>
<td>JKL</td>
<td>945</td>
<td>345</td>
<td>-75</td>
</tr>
<tr>
<td>MNO</td>
<td>-345</td>
<td>-567</td>
<td>-234</td>
</tr>
<tr>
<td>PQR</td>
<td>934</td>
<td>49</td>
<td>89</td>
</tr>
<tr>
<td>STU</td>
<td>-345</td>
<td>-945</td>
<td>789</td>
</tr>
<tr>
<td>VWX</td>
<td>34</td>
<td>-456</td>
<td>-567</td>
</tr>
</tbody>
</table>
EXTRACT RECORDS FROM A VERY LARGE DATASET

A very useful and clever way to use a format is to define the subset of a large dataset that you want to extract and read in the data using that format. It is similar to the first example in that it does define the data by groups. One group is the data that you want to keep, and the other group is the data to be discarded. But it is different in that it is used on input whereas formats are usually applied for output display.

The first part of the program uses a list of id values from a small file to identify the records that are to be extracted from the big file. ID1 is the variable we want to key on and it is used to create a format called MFMT. The CNTLIN option is used to store the format in FMTFILE. Note: The CNTLIN option of PROC FORMAT allows us to create a format from data in a SAS dataset. It requires that we create the variables FMTNAME, START, LABEL, and TYPE (for character data).

Then the big file is set in and we match the id variable there (LN_ID) to our key variable in the format MFMT. By using the subsetting IF, we instruct the program to keep only those records that match the values defined in the format. So that means we will only read in the records that match to the values from the list of ID1 values.

This technique is especially useful if the big dataset is not sorted because it does not require a sort and therefore saves the resources. It is also especially effective when the small dataset is significantly smaller than the large one.

Example 8 - Program

*--------------------------------------------------*
* Use id1 from small file to create a format   *
*--------------------------------------------------*
data keydata;
  set home.dataset8(keep=id1);
run;

data fmtfile;
  set keydata end=eof;
  length start $10
    label $10
    key $10
    ;
  key=id1;
  start=key;

  fmtname = 'mfmt';
  type = 'C';
  label = 'Match';
  output;

  if eof then do;
    start = 'other';
    label = 'No Match';
    output;
  end;
run;

proc sort data=fmtfile nodupkey;
  by key;
run;

proc format cntlin=fmtfile;
run;

---

8 – EXTRACT RECORDS FROM A VERY LARGE DATASET

A very useful and clever way to use a format is to define the subset of a large dataset that you want to extract and read in the data using that format. It is similar to the first example in that it does define the data by groups. One group is the data that you want to keep, and the other group is the data to be discarded. But it is different in that it is used on input whereas formats are usually applied for output display.

The first part of the program uses a list of id values from a small file to identify the records that are to be extracted from the big file. ID1 is the variable we want to key on and it is used to create a format called MFMT. The CNTLIN option is used to store the format in FMTFILE. Note: The CNTLIN option of PROC FORMAT allows us to create a format from data in a SAS dataset. It requires that we create the variables FMTNAME, START, LABEL, and TYPE (for character data).

Then the big file is set in and we match the id variable there (LN_ID) to our key variable in the format MFMT. By using the subsetting IF, we instruct the program to keep only those records that match the values defined in the format. So that means we will only read in the records that match to the values from the list of ID1 values.

This technique is especially useful if the big dataset is not sorted because it does not require a sort and therefore saves the resources. It is also especially effective when the small dataset is significantly smaller than the large one.

Example 8 - Program

*--------------------------------------------------*
* Use id1 from small file to create a format   *
*--------------------------------------------------*
data keydata;
  set home.dataset8(keep=id1);
run;

data fmtfile;
  set keydata end=eof;
  length start $10
    label $10
    key $10
    ;
  key=id1;
  start=key;

  fmtname = 'mfmt';
  type = 'C';
  label = 'Match';
  output;

  if eof then do;
    start = 'other';
    label = 'No Match';
    output;
  end;
run;

proc sort data=fmtfile nodupkey;
  by key;
run;

proc format cntlin=fmtfile;
run;
data address;
  drop key keyfmt;
  length key $10
  keyfmt $10
  ;
  set appldir.address;
  key = ln_id;
  keyfmt = put(key,$mfmt.);
  if keyfmt = 'Match';
run;

*-------------------*
*  Merge  *
*-------------------*
proc sort data=home.dataset8
  out=dataset8;
  by id1;run;
proc sort data=address(rename=(ln_id=id1));
  by id1;run;
data newfile(label="Dataset8 with Addresses");
  merge dataset8(in=in8) address;
  by id1;
  if in8;
run;
proc contents data=newfile;
run;

Example 8 - Log
.
.
.
38  proc sort data=fmtfile nodupkey;
40    by key;
41  run;

NOTE: 1 observations with duplicate key values were deleted.
NOTE: There were 10001 observations read from the dataset WORK.FMTFILE.
NOTE: The data set WORK.FMTFILE has 10000 observations and 6 variables.
NOTE: PROCEDURE SORT used:
      real time 0.05 seconds
      cpu time  0.04 seconds

42  proc format cntlin=fmtfile;
NOTE: Format $MFMT has been output.
44  run;

NOTE: PROCEDURE FORMAT used:
      real time 1.29 seconds
      cpu time  0.19 seconds

NOTE: There were 10000 observations read from the dataset WORK.FMTFILE.
*--------------------------------------------------*
* Subset Big file using id1 format *
*--------------------------------------------------*
data address;
  drop key keyfmt;
  length key $10
  keyfmt $10
  ;
  set appldir.address;
  key = ln_id;
  keyfmt = put(key,$mfmt.);
  if keyfmt = 'Match';
  run;

NOTE: There were 34410649 observations read from the dataset APPLDIR.ADDRESS.
NOTE: The data set WORK.ADDRESS has 9995 observations and 5 variables.
NOTE: DATA statement used:
  real time            3:11.01
  cpu time            1:37.85

CONCLUSION

PROC FORMAT is often a more efficient way of transforming data than using DATA step code. Some of the advantages to using PROC FORMAT for data manipulation are:

- It is easier to code, read, and maintain.
- It can be stored permanently.
- It can be stored separately and changed without requiring program changes.
- Values can be specified as discrete values or ranges.
- Ranges can be explicit or implicit (LOW, HIGH, OTHER)
- Formatted lookups use a binary search.
- The PUT function with a format is more CPU efficient than a MERGE statement.
- Using one PUT function is more efficient than executing multiple IF statements.
- The entire format is loaded into memory.

These examples are just a few of the many applications of PROC FORMAT to input, manipulate, and output data. With some thought and creativity, it can be used to manage data far beyond the scope of just an attractive display. I encourage you to think of PROC FORMAT more often and to devise even more interesting applications.

CONTACT INFORMATION

The author can be reached at Lois831@hotmail.com.

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.